Файфель Б.Л.

COM в действии - I. Создаем AciveX-DLL

1. Введение

Объекты и классы не сходят со страниц книг по программированию. Читателям внушаются мысли о большой пользе, которую может принести использование объектно-ориентированных подходов. Вместе с тем, такие популярные среды разработки, как VB и Delphi вполне позволяют писать достаточно серьезные приложения без явного привлечения объектной техники. У программистов, наработавших некоторый опыт, невольно возникает вопрос: а зачем нужны объекты, если я без них прекрасно обхожусь? А действительно, зачем? Заглянем в серьезные книги, и убедимся, что реальная польза от объектов достигается в больших проектах, реализующих сложные модели (серьезные графические пакеты, моделирование реальных процессов, операционные системы и тому подобное). Обычные повседневные задачи типичного программиста, скажем так, несколько скромнее перечисленных выше. Значит обойдемся без объектов? 

Не будем торопится с выводами. Цель этой заметки (рассчитанной, как читатель уже понял, на начинающего программиста) состоит в объяснении того, какую пользу может извлечь из объектного подхода даже новичок. 
Сначала давайте сформулируем две проблемы, с которыми, без сомнения, сталкивается любой серьезный программист.
Первая проблема: создание приложений, части которых (скажем, отдельные процедуры) написаны на разных языках программирования.
Допустим, мы разрабатываем приложение, работающее с базой данных и подвергающее эти данные серьезной вычислительной обработке, то было бы удобно:

· сложные вычисления реализовать на C++ (эффективный код);

· внешний интерфейс и доступ к СУБД реализовать на VB (легкость реализации).

Вторая проблема - создание библиотек программ. Вы, предположим, разработали ряд полезных процедур, и готовы ими поделиться. Но не хотите при этом передавать исходные тексты. Как быть? Построить библиотеку  процедур. Работающие на Delphi и C++ имеют возможность строить процедурные библиотеки (т.н. процедурные Dll), а любители Visual Basic лишены такой возможности. 
Существует общий подход, позволяющий решить обе приведенные проблемы. Этот подход называется "COM" (сокращение от Component Object Model - компонентая модель объектов). Строгое описание COM - слишком серьезная тема. Мы подойдем к делу чисто практически: покажем на простом примере, как создаются COM-библиотеки в среде VB, и как можно использовать созданные библиотеки в других средах программирования. 

Говоря очень упрощенно, для использования COM необходимо представить свой код в виде набора одного или более классов. 

Любой, пишущий на VB программист использует классы ("разговаривает прозой"), может быть не подозревая об этом. Из классов создаются формы и элементы управления. Эти классы можно назвать "встроенными" - их создает исполнительная среда VB. О том, как создавать свои классы (которыми можно пользоваться внутри проекта) - см. отличную статью Б. Рудого "Обьекты в Visual Basic для начинающих (коротко о создании, применении, свойствах)".

Цель этой статьи - рассказать о том, как создаются классы, которые можно скомпилировать в библиотеку и использовать где угодно: в другом VB-проекте; в VBA-приложении в составе Word, Excel, PowerPoint, в приложении на VBScript/JScript, на Web-сервере IIS, в приложениях Delphi, С++ Builder и т.д.

Такие библиотеки называются ActiveX-Dll. Кроме классов в ActiveX-Dll можно включать формы, модули, ресурсы - все, что угодно. Но чтобы воспользоваться всем этим богатством, нужно создать объект и вызывать его методы (или использовать свойства). 

2. Приступаем...

Теперь давайте займемся делом - создадим COM-библиотеку, в которой будет храниться примитивный калькулятор, выполняющий четыре действия арифметики. Библиотеку мы создадим на VB, а пользоваться ею будем в VBScript, VBA и Delphi. 

Итак, начинаем новый VB-проект. Выбираем тип "ActiveX-DLL". Сразу же переименуем проект из "Project1" в "Calc". Модуль класса "Class1", уже содержащийся в проекте, переименуем в "myCalc". Добавим в проект форму, и назовем ее "frmMain". Если все сделано правильно, то проект должен выглядеть так (Рис.1):

Рис.1
Начнем с самого простого - с формы. Разместим на форме два текстовых поля ввода, а слева от них метки с текстом "A=" и "B=". Это будут области ввода операндов. Между областями ввода разместим четыре кнопки со знаками арифметических операций ("+"," -" , "*" и "/"). Под вторым полем разместим метку с размерами, достаточными для отображения 7-10 цифр. Здесь будет отображаться результат операции. И, наконец, в нижней части разместим кнопку с надписью "Закрыть" для удаления калькулятора с экрана. У формы уберем все органы управления, выставив BorderStyle=0 и зададим размещение на экране - ручное (StartUpPozition=0). Все вместе может выглядеть примерно так (Рис. 2):

Рис. 2
Добавим в область кода формы следующие процедуры:

Private Sub Command1_Click()

        AA# = Val(Me.Text1.Text)

        BB# = Val(Me.Text2.Text)

        CC# = AA# + BB#

        Me.Label3.Caption = CStr(CC#)

End Sub

Private Sub Command2_Click()

        AA# = Val(Me.Text1.Text)

        BB# = Val(Me.Text2.Text)

        CC# = AA# - BB#

        Me.Label3.Caption = CStr(CC#)

End Sub

Private Sub Command3_Click()

        AA# = Val(Me.Text1.Text)

        BB# = Val(Me.Text2.Text)

        CC# = AA# * BB#

        Me.Label3.Caption = CStr(CC#)

End Sub

Private Sub Command4_Click()

        AA# = Val(Me.Text1.Text)

        BB# = Val(Me.Text2.Text)

       If Abs(BB#) > 0.00000001 Then

           CC# = AA# / BB#

          Me.Label3.Caption = CStr(CC#)

       Else

          Me.Label3.Caption ="Деление на нуль!"

       End if

End Sub
Private Sub Command5_Click()

        Me.Hide

End Sub

Процедуры не требуют особых комментариев: по нажатии кнопки со знаком операции извлекаем введенные данные из областей ввода аргументов, переводим в формат Double, выполняем операцию, результат переводим в строковое представление и присваиваем свойству Caption метки результата.

Формочка, которую мы нарисовали, предназначена для отображения на экране. А в каком месте экрана она будет появляться? Это зависит от того, кто будет пользоваться нашей ActiveX-Dll. Пользователь может захотеть отобразить форму калькулятора в любом месте экрана. Давайте предусмотрим возможность задать координаты левого верхнего угла нашей формы. Поскольку форма имеет фиксированные размеры, координаты левого верхнего угла определяют положение формы вполне однозначно. А если пользователь захочет отцентрировать форму в какой-либо области? В этом случае ему понадобится ширина и высота нашей формы.

В обычном проекте VB мы могли бы просто написать:

frmMain.Left=...

frmMain.Top=...

...=frmMain.Width

...=frmMain.Height

Разрабатывая ActiveX-Dll, мы не можем так поступить. ActiveX-Dll используется внешним приложением (написанным, может быть, не на VB). Это внешнее приложение не может вызывать свойства и методы нашей формы frmMain напрямую. Разработчик ActiveX-Dll должен написать некий промежуточный код, с которым будет взаимодействовать пользователь Dll, и который будет обращаться ко всей "внутренней" начинке ActiveX-Dll. Этот код должен быть оформлен в виде класса. Вы никогда не создавали классов? Ничего страшного - сейчас увидите, что это не так уж сложно.

Автор еще раз отсылает читателя к статье Б. Рудого и к специальным руководствам за более подробными сведениями о классах и объектах. Нам же сейчас достаточно представления, что:

· класс – это тип, создаваемый программистом;

· объект – это переменная такого типа. Говорят, что объект создается из класса.
· объект объединяет “под одной крышей” методы (процедуры) и свойств (переменные).
Мы пишем класс для нашего калькулятора. Из этого класса потом будет создаваться объект. Мы хотим, чтобы у класса было четыре свойства:

- Top 
 - ордината левого верхнего угла;

- Left
 - абсцисса левого верхнего угла;

- Width
 - ширина формы;

- Height
 - высота формы

Смело открываем пустой модуль класса myCalc.cls и пишем с первой строки:

':::::::::: Локальные переменные

Private mTop       As Long

Private mLeft       As Long

Private mWidth   As Long

Private mHeight  As Long

Это - место для хранения свойств.  Пользователь будет получать доступ к этим свойствам через процедуры свойств. Чтобы создать эти процедуры в меню "Tools" выбираем "Add Procedure" (Рис.3):

Рис 3
Появится окно добавления процедуры (Рис 4):

Рис 4
В поле "Name" пишем "Top", нажимаем радиокнопку "Property" (свойство), и нажимаем OK. В области кода появляются две заготовки:

Public Property Get Top() As Variant

End Property

Public Property Let Top(ByVal vNewValue As Variant)

End Property

Кто не знает, это процедуры, которые реализуют свойства: "Property Get" обеспечивает получение свойства, "Property Let" - присвоение свойству значения.Другими словами, когда пользователь нашего класса пишет где-либо:

a=Имя_Объекта.Top

то вызывается "Property Get", а когда выполняется оператор:

Имя_Объекта.Top=b

то вызывается "Property Let".

Для хранения экранных координат удобно использовать длинные целые. Исправим в заготовках свойств тип данных с Variant на Long. Аналогично создадим свойства Left, Width и Height. Остановимся на минуту. Всем хорошо известно, что бывают свойства для чтения и записи, а бывают - только для чтения или только для записи. Пара свойств Top и Left могут и читаться и модифицироваться пользователем класса. А вот свойства Width и Height имеет смысл только читать. Мы ведь установили у формы фиксированные размеры и запретили ресайзирование. Как сделать свойства Width и Height свойствами "только для чтения" (Read Only)? Да очень просто - удалить из кода соответствующие процедуры "Let"!

Что еще необходимо добавить? Код инициализации. Это процедура, которая выполняется всякий раз, когда на основе класса создается объект. Для тех, кто читал книги по объектно-ориентированному программированию могу уточнить, что код инициализации - это аналог конструктора класса. 

В процедуре инициализации самое место присвоить начальные значения всем внутренним переменным:

Private Sub Class_Initialize()

        mTop = 0

        mLeft = 0 

        mWidth = frmMain.Width

        mHeight = frmMain.Height

End Sub 

Обратите внимание на то, что процедура Class_Initialize объявлена с атрибутом "Private". Так обстоит дело потому, что эта процедура вызывается автоматически и не может быть вызвана явно. 
А кто же будет отображать нашу форму? Кто вызовет метод Show? Нам нужно создать в коде класса Public-процедуру; ее можно так и назвать:

'::::::::::: Отобразить окно

Public Sub Show()

       frmMain.Show 1

End Sub 

Обратите внимание, что форма отображается как модальная.

Теперь заполним пустые места в заготовках вполне очевидным образом:

Public Property Get Left() As Long

       Left = mLeft / Screen.TwipsPerPixelX

End Property

Public Property Let Left(ByVal vNewValue As Long)

       mLeft = vNewValue * Screen.TwipsPerPixelX

       frmMain.Left = mLeft

End Property

Public Property Get Top() As Long

       Top = mTop / Screen.TwipsPerPixelY

End Property

Public Property Let Top(ByVal vNewValue As Long)

       mTop = vNewValue * Screen.TwipsPerPixelY

       frmMain.Top = mTop

End Property

Public Property Get Height() As Long

       Height = mHeight / Screen.TwipsPerPixelY

End Property

Public Property Get Width() As Long

       Width = mWidth / Screen.TwipsPerPixelX

End Property

Обратите внимание: геометрические характеристики хранятся в твипах, а принимаются и возвращаются процедурами-свойствами в пикселах. Так пришлось поступить потому, что твипы могут не поддерживаться в средах, отличных от VB, а мы строим универсальную библиотеку. Для пересчета пикселов в твипы мы используем величины Screen.TwipsPerPixelX (пересчет горизонтальных размеров) и Screen.TwipsPerPixelY (пересчет вертикальных размеров).
Процедуры Get просто возвращают значения, хранящиеся в локальных переменных. Процедуры Let меняют значения этих переменных на значения своих параметров и одновременно устанавливают соответствующие свойства формы.
Когда из нашего класса создается объект, создается и форма frmMain, но остаетя невидимой. Создав объект, пользователь получает возможность установить свойства "Left" и "Top", а потом, вызвав метод Show (не формы, а класса!) отобразить форму на экране. 
Класс практически готов. Давайте добавим к нему еще один метод: вычисление из строки. Имеется в виду следующее. Мы создаем мощнейший калькулятор с визуальным интерфейсом (шутка!) А вдруг кому-нибудь понадобиться вычислить что-либо, не отображая форму, а просто передав объекту строку параметров типа:  A,B,"+" ? Вот для этого и пригодится следующий код:

'::::::::::: Вызвать вычислитель без отображения формы

Public Function calculate(A As Double, B As Double, action As String) As Double

       Select Case (action)

              Case "+"

                  calculate = A + B

              Case "-"

                  calculate = A - B

              Case "*"

                  calculate = A * B

              Case "/"

                  calculate = A / B

       End Select

End Function 

Как хорошо видно, у этого метода три входных параметра - два операнда и код операции (в виде строки). А возвращает этот метод результат операции.

Сейчас все готово для трансляции. Транслируем (выбрав "File" -> make Calc.dll) У нас в текущей директории образуется ActiveX-Dll с именем Calc.dll. 

3. Используем...

Теперь будем пробовать работоспособность нашей библиотеки. Начнем с самого простого - с VBScript.

Создаем текстовым редактором (только не Word-ом!) текстовый файл test_Calc.Vbs:

Set C=CreateObject("Calc.myCalc")

C.left=200

C.Top=200

C.show

Обратите внимание на первую строку: это создание объекта из ActiveX-Dll. Имя dll указывается до точки, а после точки - имя класса. А почему не просто: Set c=CreateObject("Calc") ? А потому, что одна Dll может хранить много классов с разными именами. Так что все очень логично. Переменная "C" хранит объектную ссылку. Для действия с объектными ссылками предназначен оператор Set.

Три последующих строки нам понятны: устанавливаем координаты левого верхнего угла формы и отбражаем ее (Рис. 5):

Рис 5
Невероятно, но факт:  работает...

Давайте попробуем воспользоваться нашей ActiveX-Dll в Visual Basic. Создаем новый проект (на этот раз - стандартный EXE). Добавляем в него форму, а на форму сажаем три кнопки (Рис 6):

Рис 6
Процедуру события для первой кнопки ("Показать") зададим такой:

Private Sub Command1_Click()

Dim Ca As Object

    Set Ca = CreateObject("Calc.myCalc")

    Ca.Left=(Me.Left+(Me.Width-Ca.Width*Screen.TwipsPerPixelX)*0.5)/ creen.TwipsPerPixelX

    Ca.Top=(Me.Top+(Me.Height-Сa.Height*Screen.TwipsPerPixelY)*0.5)/Screen.TwipsPerPixelY

    Ca.Show

End Sub

В процедуру события для третьей кнопки ("Закрыть") занесем оператор End. Процедуру второй кнопки пока задавать не будем.

Легко видеть, что громоздкие выражения в приведенной выше процедуре - не что иное, как центрирование формочки-калькулятора на главной форме нашего проекта. 
Обратите внимание, что функция CreateObject  - создания объекта из ActiveX-Dll имеет тот же синтексис, что и в VbScript.

Запустим проект и нажмем кнопку "Показать". Калькулятор, как и следовало, появится в середине формы (Рис 7):

Рис 7
Ну, а как насчет использования нашей ActiveX-Dll в Excel и Word? Автор, помнится, обещал это в начале статьи... И автор свое слово держит!

Создаем рабочую книгу Excel. Заходим в редактор VBA (нажав Alt-F11) и в рабочей области книги пишем... совершенно верно, именно это:

Sub myCalc()

Dim C As Object

    Set C = CreateObject("Calc.myCalc")

    C.Left = 200

    C.Top = 200

    C.Show

End Sub

Чтобы проверить работоспособность процедуры, вернемся на рабочий лист,  нажмем Alt-F8 (Рис 8):

Рис 8

а затем - кнопку "выполнить". Убедимся, что наш супер-калькулятор появится на экране (Рис 9):

Рис 9
Совершенно аналогично дело обстоит и в Word. Автор решил не приводить картинки-подтверждения. 
А как Вы думаете, уважаемый читатель, откуда Excel, Word, VB знают, где искать библиотеку Calc с классом myCalc внутри? Мы только что скомпилировали calc.dll. Давайте скопируем calc.dll в другую директорию. Скопировали? А теперь запускаем наш первый скрипт test_calc.vbs. Опа! Ошибка (Рис 10):

A, это понятно - скажет читатель. Библиотеки нет в текущей директории, вот скрипт и валится. Ладно, копируем скрипт в ту же дириекторию, где сидит calc.dll. Запускаем... Что, опять ошибка? А теперь копируем calc.dll обратно. Запускаем скрипт. Работает. Мистика какая-то! Что, система помнит, в какой директории библиотека компилировалась и запускает ее только из "родной" директории? 

4. Регистрация DLL

Ну, хватит мистики! В реальности дело обстоит так:

Каждая ActiveX-Dll должна быть ЗАРЕГИСТРИРОВАНА на компьютере. В чем состоит суть регистрации? В том, что создается запись в системном реестре (ну, куда же без реестра!). В этой записи фиксируется, в частности, директория, где  Dll проходила регистрацию. А зачем так сложно? А затем, что теперь любой исполняемый модуль или скрипт может пользоваться этой библиотекой, не заботясь о ее нахождении в текущей директории. Не верите? Скопируйте скрипт test_calc.vbs на любой диск Вашего ПК (да хоть на дискету!). И запустите. Работает? То-то.

Но позвольте, а мы Dll только скомпилировали. Кто же выполнил регистрацию? Visual Basic - больше некому! Когда мы компилируем ActiveX-Dll, Visual Basic ee автоматически регистрирует в директории создания. Если же мы хотим передать эту Dll на другой ПК, то регистрацию придется делать новому хозяину Dll. Зарегистрировать Dll просто: копируем ее в заданную директорию (скажем в C:\Windows\System32) и, находясь в этой директории, исполняем команду:

regsvr32.exe Имя_Dll

Благополучное завершение регистрации выглядит так (Рис 11):

Рис 11
Таким образом мы с Вами, уважаемый читатель, создали вполне полноценную COM-библиотеку. Она содержит единственный класс. А могла бы содержать десять или двадцать классов (сколько потребуется). Чтобы добавить к существующему проекту ActiveX-Dll еще один класс, достаточно добавить еще одни модуль класса (а также, формы, программные модули, ресурсы) и запрограммировать необходимые методы и свойства. После этого необходимо перекомпилировать проект. На своей машине больше ничего делать не нужно, а если Вы передаете Dll кому-либо, то на другой машине придется повторить регистрацию. А почему? Ведь мы скопируем Dll на место старой, в ту же директорию... Поворить регистрацию нужно потому, что кроме директории в реестре для каждого класса из Dll запоминается специальный длинный ключ, т.н. GUID. Читатель, хоть раз заглянувший в реестр, видел эти самые GUID-ы во множестве. Хотите посмотреть, что записано в реестре для нашей Dll? Пожалуйста. Запускаем regedit, выбираем “искать” и задаем строку поиска “Calc.myCalc”. Мы увидим примерно следующее (Рис 12):

Рис 12
Как видите, “папка” InprocServer32 “открыта”. Ее содержимое приводится ниже (Рис 13):

Рис 13
Теперь ясно, где хранится путь к зарегистрированному экземпляру Dll. На самом деле для каждой зарегистрированной Dll в реестре создается несколько записей. Но эти подробности для нас сейчас не важны; желающие могут взяться за серьезное изучение COM. Несколько хороших книг на эту тему – в конце статьи.

Имейте в виду, что одну и ту же Dll можно на одной машине зарегистрировать в разных директориях. При регистрации каждый раз будет генерироваться новый ключ-GUID. 

5. Раннее и позднее связывание

Теперь обратим внимание на один очень важный момент. Когда мы работаем в среде VB/VBA со стандартными классами (формами, элементами управления), среда программирования откуда-то знает весь набор свойств и методов класса. Вспомним, когда мы ставим точку после имени формы, появляется список свойст и методов (Рис 14):

Рис 14
А чем наш класс myCalc хуже? Давайте вернемся к нашему VB-проекту и в процедуре Command1_Click после строки Set Ca=CreateObject(...  напишем “Ca.” и подождем появления списка. Что, не появляется? Значит, наш класс неполноценный? Нет, класс вполне полноценный. Просто мы им “не вполне полноценно” пользуемся. Давайте попробуем понять, почему это так.

Информация о свойствах и методах класса, очевидно, содержится в самой Dll. Программе этот список и методов становится доступным только после того, как будет создан объект (выполнится оператор Set Ca=CreateObject(...)  т.е. уже во время выполнения). А во время разработки IDE Visual Basic еще не подозревает о существовании нашей Dll. Такой метод создания объектов называется поздним связыванием. А чтобы сообщить IDE Visual Basic (и компилятору!) о существовании нашей замечательной Dll нужно установить не нее ссылку. 

Давайте создадим копию нашего VB-проекта в другой директории, зайдем в свойства проекта (Рис 15):
Рис 15
и щелкнем по пункту “References” (“Ссылки”). На экране появится следующая форма (Рис 16):

[image: image22.png]


Рис 16
Листаем список вниз, пока не найдем нашу Dll (она будет называться Calc), а затем возводим флажок слева от имени и нажимаем OK. Ссылка установлена.

Использование Dll с предварительной установкой ссылки называется ранним связыванием – теперь IDE Visual Basic (и компилятор) “знают” полный список свойств и методов класса. Кстати, по терминологии COM, совокупность свойств и методов класса называется интерфейсом класса.  

Откроем теперь окно кода  и заменим описание Сa в процедуре Command1_Click на такое: Dim Ca As calc.myCalc.   Ниже этой строки пишем “Ca.” - о, чудо!  - список появился (Рис 17):

Рис 17
Теперь немного изменим процедуру Command1_Click:

Private Sub Command1_Click()

Dim Ca As calc.myCalc

    Set Ca = New calc.myCalc

    Ca.Left=(Me.Left+(Me.Width-Ca.Width*Screen.TwipsPerPixelX)*0.5)/ creen.TwipsPerPixelX

    Ca.Top=(Me.Top+(Me.Height-Сa.Height*Screen.TwipsPerPixelY)*0.5)/Screen.TwipsPerPixelY

    Ca.Show

End Sub
Убедимся, что новый проект работает не хуже старого: окно калькулятора появляется, где положено, а сам калькулятор исправно работает. Следует ли отсюда, что раннее связвание полезно только тем, что ассистент VB подсказывает свойства и методы класса? Разумеется, нет! Есть еще более веская причина использовать ранее связывание. Чтобы убедиться в этом, вернемся сначала в VB-проект с поздним связыванием и добавим следующий код в обработчик события Command2_Click (помните, мы оставили вторую кнопку без обработчика):

Private Sub Command2_Click()

Dim Ca As Object

    Set Ca = CreateObject("Calc.myCalc")

    TBeg$ = Time

    For ii& = 1 To 1000000

        r# = Ca.Calculate(50#, 60#, "*")

    Next ii&

    TEnd$ = Time

    MsgBox TBeg$ + Chr$(10) + TEnd$

End Sub

Что здесь происходит? Миллион раз вызывается метод Calculate (вот и пригодился!) и вычисляется произведение 50*60. Перед циклом и после цикла фиксируется время, которое потом и показывается пользователю. Все ясно? Запускаем. Не знаю, как у Вас, а на моем компьютере получается результат (Рис 18):

Рис 18
На вычисления, как видим, ушло примерно 15 сек. Очень хорошо. А теперь перейдем в проект с ранним связыванием, допишем обработчик для клика на второй кнопке:

Private Sub Command2_Click()

Dim Ca As calc.myCalc

    Set Ca = New calc.myCalc

    TBeg$ = Time

    For ii& = 1 To 1000000

        r# = Ca.calculate(50#, 60#, "*")

    Next ii&

    TEnd$ = Time

    MsgBox TBeg$ + Chr$(10) + TEnd$

End Sub

Код точно такой же, как и при позднем связывании, но объект описывается и создается по-другому. Запускаем (Рис 19):

Рис 19
Вот это да! Всего 1 секунда!! В 15 раз быстрее, чем при позднем связывании!!! Вот главное преимущество раннего связывания – значительное возрастание производительности. А как Вы думаете, за счет чего так получается? Давайте попробуем разобраться. Посмотрим еще раз на код:

r# = Ca.Calculate(50#, 60#, "*")

При позднем связывании компилятор “не знает”, что у класса, из которого создан объект Ca, есть метод “Calculate”. И он (компилятор) вынужден генерировать код, который при каждом вызове обращается к объекту с запросом: “у тебя есть метод Calculate?” И так миллион раз... А при раннем связывании компилятор “знает”, что у класса, из которого создан объект Ca, есть метод “Calculate”, и просто вызывает его! 

6. Какое связывание лучше?

Так что же, забудем о позднем связывании, раз ранее связывание так удобно? Не будем торопиться с окончательным заключением – у позднего связывания есть примущества (а у раннего – недостаки). Поздним связыванием проще пользоваться (кроме того, в cкриптовых языках типа VBScript только поздним связыванием и можно пользоваться). Теперь предположим, что мы в очередной раз перекомпилировали Dll. Пользователь Dll должен ее перерегистрировать (и при позднем, и при раннем связывании). Но при позднем связывании больше ничего делать не нужно. А вот при раннем связывании нужно будет переустановить ссылку и перетранслировать программу, которая использует нашу Dll. Если просто запустить код, который использует старую Dll, то Вы получите примерно такое сообщение: (Рис 20)

Рис 20
Окончательный выбор за Вами, уважаемый читатель! Но автор может дать пару простых советов:

1) Если Вы разработали Dll и твердо решили не менять ее – раннее связывание предпочтителнее – удобнее и быстрее!

2) Если Вы  планируете менять версии – удобнее позднее связывание. 
Чтобы повысить производительность программ (при позднем связывании) имейте в виду следующее: самая долгая операция – вызов метода. Но когда управление получает код, содержащийся в Dll, то он выполняется так же быстро, как и в обычном exe-файле. Отсюда вывод: проектируйте свои Dll так, чтобы вызвав метод один раз, получить все, что требуется (избегайте большого количества вызовов при позднем связывании). В этом случае, Вы сможете предотвратить критическое снижение производительности.

Так... Что автор забыл сказать? Как активизировать раннее связывание в Word-е и Excel-e. Так же просто, как и в VB. Заходим в редактор VBA и выбираем пункт меню “Сервис” (“Tools”). Первый же подпункт позволяет задать ссылки для VBA-проекта. Даже иконка такая же, как у VB. 

7. А как насчет других языков?

Ехидный  читатель в этом месте может сказать: “Ну, ладно, Vb, Vba, VBscript... В конце концов, это все Microsoft-овские языки. А как начет, например, Delphi?” 

Дальнейшее содержание адресовано тем, кто наряду с VB пишет и на Delphi. Вы убедитесь, ActiveX-Dll, сделанные в VB, прекрасно работают и в Delphi.

Теперь, когда мы знаем про два типа связывания, посмотрим, как они реализуются в среде Delphi. Начнем с позднего связывания. 

Создаем новый проект на Delphi (у меня Delphi-4). Размещаем на форме три кнопки (Рис 21):

Рис 21
К списку директивы Uses добавим ссылку на ComObj. И зададим обработчики кликов:

procedure TForm1.Button1Click(Sender: TObject);

Var CLC    : Variant;

Var XX,YY : Word;

begin

    CLC:=CreateOLEObject('Calc.myCalc');

    XX:=Form1.Left +(Form1.Width-CLC.Width)*0.5;

    YY:=Form1.Top +(Form1.Height-CLC.Height)*0.5;

    CLC.Top:=YY;

    ClC.Left:=XX;

    CLC.Show;

end;

procedure TForm1.Button2Click(Sender: TObject);

Var i                   : integer;

Var CLC             : Variant;

Var Tbeg,Tend  : String;

begin

    CLC:=CreateOLEObject('Calc.myCalc');

    Tbeg:=TimeToStr(Time);

    For i:=1 to 1000000 Do

        CLC.Calculate(100,200,'+');

    TEnd:=TimeToStr(Time);

    ShowMessage('Tbeg='+Tbeg+chr(10)+'Tend='+Tend);

end;

procedure TForm1.Button3Click(Sender: TObject);

begin

     Application.Terminate; 

end;

Вы немного разочарованы? Все то же самое, что и в VB (с учетом разницы в синтаксисе). Функция CreateObject в Delphi называется CreateOLEObject. Для хранения ссылок на объекты, создаваемые из Dll (OLE-объекты) служит специальный тип данных Variant (в VB - Object). Обратите внимание на то, как просто центрируется форма калькулятора на главной форме - все в пикселах! Запускаем проект, и убеждаемся, что все работает. Неприятный сюрприз состоит в том, что вычисления длятся почти вдвое дольше, чем в VB. Почему? Видимо, вызов метода при позднем связывании из Delphi еще накладнее, чем из VB.

Ну, а раннее связывание? С ним чуть сложнее. Автор вынужден ввести еще одно определение. Формализованное описание интерфейса COM-объекта (помните, интерфейс - это совокупность методов и свойств) называется библиотекой типов. Для создания библиотек типов существует специальный язык IDL. К счастью нам, VB-программистам, нет необходимости его изучать. О нас позаботилась Microsoft: наши ActiveX-Dll включают библиотеки типов. Для того, чтобы осуществить раннее связывание нам нужно (в терминологии Delphi) осуществить импорт библиотеки типов. Создадим копию проекта в другой директории, выберем пункт меню "Project" (Рис 22):

Рис 22

и далее "Import Type Library". Далее появится список зарегистрированных объектов (Рис 23):

Рис 23
Находим наш калькулятор, и нажимаем кнопку "OK". К проекту добавляется новый Unit внушительных размеров. Имя Unit строится из имени Dll, дополненной конструкцией "_TLB"  Желающие могут изучить Unit. А мы пойдем дальше. Установим в Unit1 ссылку на calc_TLB. Добавим в область переменных Unit описание нашего объекта:

Var  clc  : myCalc;

Зададим обработчики:

procedure TForm1.Button1Click(Sender: TObject);

begin

          CLC.Top :=Form1.Top+(Form1.Height-clc.Height) div 2;

          CLC.Left:=Form1.Left+(Form1.Width-clc.Width) div 2;

          CLC.Show;

end;

procedure TForm1.FormCreate(Sender: TObject);

begin

      CLC:=coMyCalc.Create;

end;

procedure TForm1.Button2Click(Sender: TObject);

Var i                  : integer;

Var Tbeg,Tend : String;

Var A,B,C          : Double;

Var Op              : widestring;

begin

    A:=100;

    B:=200;

    Op:='+';

    Tbeg:=TimeToStr(Time);

    For i:=1 to 1000000 Do

        C:=CLC.Calculate(A,B,Op);

    TEnd:=TimeToStr(Time);

    ShowMessage('Tbeg='+Tbeg+chr(10)+'Tend='+Tend);

end;

В процессе набора кода обратите внимание на то, что ассистент Delphi тоже показывает список свойств и методов объекта (ну, может быть, не так симпатично, как VB). Создается объект оператором (выполняющимся в обработчике события создания формы):

CLC:=coMyCalc.Create;

И еще. При передаче строковых параметров нужно употреблять тип widestring (строка Unicode). Остальное ясно. Запустим вычисления, и убедимся, что они выполняются даже несколько быстрее, чем в VB.

В следующей статье мы рассмотрим построение реального компонента для IIS, обеспечивающего закачку файлов на сервер (т.н. Uploading).

Все приведенные примеры можно скачать здесь: 

В заключение, как и обещал автор, небольшой список литературы:

1. Брюс Мак-Кинни "Visual Basic - крепкий орешек" Microsoft Press М. 1998

Эта книга - без преувеличения - настоящий шедевр!
2. Эш Рофэйл, Яссер Шохауд "COM и COM+ полное руководство" М. 2000

3.  Visual Basic 6 Distributed экзамен 70-175 М. 2001

